Math 276
Calculus III

Dr. Constant J. Goutziers
Department of Mathematics, Computer Science and Statistics
goutzicj@oneonta.edu

Lesson 13
Tangent Planes and Differentials

Initializations

```
> restart;
  with(liesymm):
>
```

13.1 Tangent Planes

The tangent plane at the point $(x_0, y_0, f(x_0, y_0))$ to the surface $z = f(x, y)$ is given by

$$z - f(x_0, y_0) = f_x(x_0, y_0)\,(x - x_0) + f_y(x_0, y_0)\,(y - y_0)$$

Examples

Example 13.1.1

Find the equation of the tangent plane at the point with *x*-coordinate 2 and *y*-coordinate 5 to the surface

$$z = f(x, y) = x^2 y^3 + 5\,x\,y + 4$$

Solution

Use the result stated above.

```
> f:=(x, y)->x^2*y^3+5*x*y+4;
  TP:=z-f(2, 5)=D[1](f)(2, 5)*(x-2)+D[2](f)(2, 5)*(y-5);
```

$$f := (x, y) \to x^2 y^3 + 5\,x\,y + 4$$

$$TP := z - 554 = 525\,x - 2600 + 310\,y \qquad \textbf{(2.1.1.1)}$$

The formula above can be beautified by using Maple's **sort** routine.

```
> ans:=sort(rhs(TP)-lhs(TP)=0, [x,y,z]);
```

$$ans := 525\,x + 310\,y - z - 2046 = 0 \qquad \textbf{(2.1.1.2)}$$

```
>
```

13.2 Differentials

Maple's routine **d** for working with differentials is located in the **liesymm** package which was loaded in the **Initializations** section of this worksheet. Before using the **d** facility we have to setup the independent variables.

```
> setup([x, y]);
```

$$[x, y] \qquad \textbf{(3.1)}$$

```
> dg:=d(g(x, y));
```

$$dg := \left(\frac{\partial}{\partial x} g(x, y) \right) d(x) + \left(\frac{\partial}{\partial y} g(x, y) \right) d(y) \qquad \textbf{(3.2)}$$

```
>
```

Examples

Example 13.2.1

Use differentials to estimate the amount of metal in a closed cylindrical can that is 10cm high and 4cm in diameter, if the metal in the wall is 0.05 cm thick and the metal in the top and bottom is 0.1 cm thick.

Solution

We first produce the formula for the volume of the can as a function of the radius *r* and the height *h*.

```
> V:=Pi*r^2*h;
```

$$V := \pi r^2 h \qquad \textbf{(3.1.1.1)}$$

Now we can set-up the variables and compute *dV*.

```
> setup([r, h]);
  dV:=d(V);
```

$$[r, h]$$

$$dV := 2 \pi r\, d(r)\, h + \pi r^2\, d(h) \qquad \textbf{(3.1.1.2)}$$

To obtain the desired answer we substitute the appropriate parameter values for *r*, *h*, *dr* and *dh* into the formula for *dV*.

```
> pars:={r=2, h=10, d(r)=0.05, d(h)=0.2};
  Est:=evalf(subs(pars, dV));
```

$$pars := \{r = 2, h = 10, d(r) = 0.05, d(h) = 0.2\}$$

$$Est := 8.796459431 \qquad \textbf{(3.1.1.3)}$$

Note: If so desired, the terms in an expression like

$$dV := 2 \pi r\, d(r)\, h + \pi r^2\, d(h)$$

can be rearranged by using the **sort** routine.

```
> sort(dV, [d(r), d(h)]);
```

$$2 \pi r h\, d(r) + \pi r^2\, d(h) \qquad \textbf{(3.1.1.4)}$$

```
>
```